
Haskellのための非同期局所化 π 計算に基づく
ネットワークプログラミングフレームワーク

今 井 敬 吾† 結 縁 祥 治† 阿 草 清 滋†

我々はプログラミング言語 Haskell へ，型付き非同期局所化 π 計算（typed Asynchronous Lo-
calized π-calculus; ALπ）を，ネットワークプログラミングのためのフレームワークとして埋め込む
手法を提案する．本フレームワークは埋め込み言語として実装されることで以下の利点を持つ：(1)
Haskellで構築されているため処理系が頑健である，(2) Haskellのリテラル，型や関数といった言語
要素をフレームワークに組み込むことができる．さらに，ALπ の mobilityに対する制限はフレーム
ワークの実装を簡単にする．本フレームワークでは，ALπ のプロセスは PiMonadと呼ばれるHaskell
のモナドとして実装する．結果として，通信により引き起こされる副作用は型の PiMonad タグで区
別される．ALπ のサブタイプ関係は Haskell の多引数型クラスで実現する．サブタイプ関係にある
型の対は通信の方向を反映した 3 つの 2 引数型クラスに属する．本フレームワークを用いた例とし
て TCP/IP ネットワーク上のインスタント・メッセンジャアプリケーションの構築例を示し，有用
性と利点を述べる．

A Network Programming Framework in Haskell
Based on Asynchronous Localized π-calculus

Keigo Imai,† Shoji Yuen† and Kiyoshi Agusa†

We propose an embedding of the typed Asynchronous Localized π-calculus (ALπ) into the
programming language Haskell as a framework for network programming. The framework
has following advantages due to the embedded language nature: (1) the framework is ro-
bust due to being built upon the Haskell framework, and (2) the framework can incorporate
various Haskell language elements, such as literals, types, and functions, in the framework.
Moreover, the limitation of mobility in ALπ simplifies the implementation of the framework.
ALπ processes are implemented by means of a Haskell monad called PiMonad. As the result,
side-effects caused by communications are distinguished by the tags of PiMonad in typing.
The subtyping relations is realized by multi-parameter type classes in Haskell, where a pair
of subtype-related types belongs to three binary type classes reflecting the directions of com-
munication. We illustrate the usefulness and benefits of our framework with an example of
an implementation of instance messenger application over TCP/IP network.

1. は じ め に

ネットワークアプリケーションの普及により，動的

な通信リンクの変化をともなうネットワークプログラ

ムはより一般的になりつつある．このようなプログラ

ムにおいては，動的なネットワーク構造を持つ柔軟な

ソフトウェアの構成が可能である反面，ソースプログ

ラムの記述だけから振舞いを十分に把握することは難

しい．

本論文ではこうした動的なネットワーク構造の変化

を記述できる体系である π 計算10) に基づいてプログ

versity

ラミング言語 Haskell（以下，単に ‘Haskell’と書く）

に対する埋め込み言語を定義することによって，ネッ

トワークプログラムの振舞いを直接記述できるフレー

ムワークを提案する．π 計算では，チャネルの送受や

生成を行う並行プロセスを柔軟に記述でき，型理論や

プロセスの等価性に関する多くの解析技法が開発され

ている．

我々は Haskell のための型付き非同期局所化 π 計

算（typed Asynchronous Localized π-calculus；以下，

ALπ と略記する）に基づくネットワークプログラミン

グフレームワーク PiMonadを実装する．ここでALπ

の型システムと構文をHaskellに埋め込み，PiMonad

プログラムのネットワーク機能の動作を Haskellで実

装する．

ALπ は，π 計算を次の 2つの点で限定した体系で

ある．(1)出力プレフィックスの後に 0 以外のプロセ

スを認めない（非同期性），(2) 名前の入力チャネル

に用いられる名前を他のプロセスに出力しない（局所

性）．非同期性は，計算能力を限定することなく，実

装を容易にする12)．局所性によって通信の入力チャネ

ルの移動が制限され，通信が発生するホストを構文的

に特定することができる．このような特徴は，IP ア

ドレスやノード番号に基づいて通信を行う多くのネッ

トワークプログラムの概念によくあてはまり，比較的

低レベルなネットワークプログラミングのモデルとし

て適当である．

PiMonadにおいて，ALπ の構文は Haskellのモナ

ドとして表現される．Haskellで入出力に用いられる

IO モナドと同様に，通信等の副作用をともなう部分

を，PiMonadとして分離して記述するのが特徴であ

る．純粋関数的部分は Haskellの処理系および標準ラ

イブラリを利用することができる利点がある．さらに，

埋め込み言語によって実装することで，Haskell処理

系が持つ厳密さに基づいて言語処理系を構築すること

ができる．

本論文では，PiMonadの構成，Haskell の I/O シ

ステムへの適用，さらにネットワーク上での実装を示

す．まず I/Oシステムとの相互作用をともなわない体

系，π 計算の記述のための関数および評価器について

定義する．型付き言語である ALπ を Haskellに埋め

込む際の問題の 1つは，Haskellの型システムにはサ

ブタイプ機構がないことである．

本論文では，サブタイプ関係を多引数型クラス

（multi-parameter type class）を用いて表現する．次

に，I/Oシステムとの相互作用をともなう体系への拡

張を示す．IO モナドを出力用チャネルとして表し，こ

の実装により一般的な入出力の記述が可能であること

を示す．さらに，ネットワーク入出力機能を表す IOモ

ナドを提供し，チャネル表現に通信ホストの位置情報

を組み込むことで，ネットワークを介した名前通信へ

拡張する．このように拡張を行うことで実際的なネッ

トワークプログラムが抽象的なモデルに近い形で直接

的に記述できる．

ALπ の利点は，ネットワーク実装において示され

る．局所性を仮定しない場合，通信がネットワーク上

のどのホストで発生するのか特定できず，実装が複雑

になる．本フレームワークは文法を ALπ に限定し，

記述の対象から複雑な挙動を除外する．このため，処

理系としては軽量な実装が可能になっている．

本論文の構成は以下のとおりである．2章で我々が

用いる ALπ の構文と操作意味論を紹介する．3章で

は非同期 π 計算に対する PiMonadを構成する方法を

示す．4章では π 計算の i/o型とサブタイプ関係によ

る局所性を型クラスで実現する方法について述べる．

5章では Haskellの入出力システムとの相互作用のた

めに拡張する．6章でネットワーク実装について述べ，

7章でプログラム例を与える．8章で関連研究につい

て述べ，9章でまとめと今後の課題について述べる．

2. 型付き非同期局所化 π 計算

本章では，本論文で用いる ALπ の構文と型システ

ムおよび意味を述べる．表現力の例としてセマフォの

記述例をあげる．

2.1 構文と基本動作意味

本フレームワークが対象としている π 計算の構文

を図 1 に示す．値は，通信リンクまたは他の基本的な

値からなる．構文に対するラベル付き遷移関係による

基本動作意味を図 2 に示す．fn(α) は α の中で束縛

されていない名前の集合，bn(α) は束縛された名前の

集合を示す．プロセス式 P に出現する自由な名前の

集合を fn(P ) とし，P における自由な z の出現の y

での置き換えを P{y/z} で表す．以上の定義に基づい
て，プロセスは以下のような直観的な意味を持つ．

• 入力：プロセス v(x).P は，名前 v を持つ出力プ

ロセスが存在する場合に値を入力し，P 中の x

を入力した値に置き換える．

• 非同期出力：プロセス vw.0 は，名前 v を持つ入

力プロセスが存在する場合に値 w を出力し，終

了する．vw と略記する．

• 並行合成：P |Q は，並行に動作する 2 つのプロ

セス P，Q を表す．

• 名前制限：(νx : L)P において，型 L を持つ名

Values

v, w ::= x channel

| basval basic value

Processes

P, Q ::= vw.0 asynchronous output

| v(x).P input

| P |Q parallel execution

| (νx : L)P restriction

| !v(x).P replicated input

| 0 terminated process

図 1 ALπ：プロセスと値の文法
Fig. 1 ALπ: grammar of processes and values.

図 2 ALπ の遷移規則
Fig. 2 ALπ: operational transition rules.

Actions

π ::= xy output action

| xy input action

| x(z) bound output

| τ internal action

図 3 ALπ：アクション
Fig. 3 ALπ: actions.

前 x のスコープを P に制限する．文脈から明ら

かな場合，型の記述を省略し (νx)P と略記する．

• 複製：!v(x).P 式は，式 v(x).P の無限の並行合

成 v(x).P |v(x).P | · · · と見なされる．
• 終了：0 式は，終了したプロセスを表す．

以下，(νx)(vx|P ) を，最も弱い結合力の演算子

v(x)P として略記する．

ラベルは図 3 で表される．xy は値 y のチャネル x

を介した出力を表す．xy は値 y の x を介した入力で

ある．x(z) は x を介した制限された名前の出力を表

す．ここで z は束縛された名前であり，スコープ射出

（scope extrusion）規則によって，他の名前と区別さ

れる．τ は他のプロセスから観測できない内部動作を

表す．

2.2 非同期性，局所性と表現能力

ALπ は，一般的な π 計算に，非同期性および局所

性の制約を加えた体系である．

2.2.1 非 同 期 性

非同期 π 計算1),6) は出力プレフィクスの継続が 0

に限られる体系である（vw.0）．一方，同期 π 計算で

は，出力プレフィクスの継続に任意のプロセスが許さ

れる（vw.P）．非同期 π 計算は TCP等の既存のネッ

トワークでの実装が簡易である利点がある．

2.2.2 局 所 性

π 計算における局所性（locality）とは，

• 入力プロセス a(x).P について，束縛された名前

xが継続する P で入力プレフィクスに出現しない，

• チャネルの同一性を検査するマッチング
[ x = y ] P の構文を認めない，

という制約である．このような体系は局所化 π 計

算8),17)（Localized π）と呼ばれる．

局所化 π 計算では，チャネルの出力能力のみをや

りとりする．入力能力は ν 演算子によってしか導入

できない．たとえばプロセス P，Q 間でチャネル x

の入力能力を共有する場合，νx(P |Q) の形をとらな

ければならない．このことは，同一の通信ホストに属

するプロセス間でのみ入力が共有できることを表して

いる．

ALπ の表現力 Milnerによる λ 計算から π 計算

へのエンコーディングが存在する9)が，この変形とし

て，ALπ へのエンコーディングも知られている8)．

他の例として，ALπ における，二値セマフォの例

をあげる．ここでは，π 計算で一般的な拡張として，

図 1 の Valuesの定義に対 (x, y) と空値 () を加えた

体系を用いる：

Semaphore(new) = !new(x).(νp, v) (x〈p, v〉
| (νa)( !a().p(r).(r〈〉|v(s).(s〈〉|a〈〉)) | a〈〉) )

ここで new はセマフォを生成するためのチャネルで

ある．new を介して得たチャネル x は生成したセマ

フォの P操作と V操作のためのチャネルをユーザプ

ロセスに渡すために用いる．チャネル a はセマフォを

初期化するために用いる．内側の a〈〉 は，V 操作の

後，新たに次の P操作を可能にする．この例は二値セ

マフォの例だが，さらに外側に 1つある a〈〉 を n 個

並行合成すれば n 値セマフォも実現できる．

ユーザプロセスにおいて，各操作はそれぞれ p(r)r()，

v(r)r() と書く．ここで r は同期のためのチャネルで

ある．たとえば，P操作の後，チャネル SomeWork

に出力し，最後に V 操作を行うユーザプロセスは以

下のように書ける：

User(new) =

new(x)x(p, v). p(r)r(). SomeWork(w)w(). v(s)s()

しかしながら局所性は，実装を容易にする一方，明

らかに表現力を損なっている．たとえば，以下のよう

なプロセスは，ALπ には属さない：

a(x).x(y).0 | νz ( az | zb )

このようなプロセスでは，制限されない並行合成演

算子を越えて，入力チャネルが移動する．このため，

入出力両方のチャネルが全体の構造上を自由に移動す

ることになる．このことは，計算の自由度が増す反面，

実際の計算資源との対応を複雑にする．

2.3 型付け規則

型のための構文を図 4 で与える．型はプロセス型と

値型，リンク型の 3つがある．リンク型は，以下のよ

うな意味を持つ．

• 入力能力 iV：入力プロセス x(y).P にのみ用い

ることができる名前の型．

• 出力能力 oV：出力プロセス xv にのみ用いるこ

とができる名前の型．

• チャネル �V：入力および出力の両方に用いるこ

とができる名前の型．

値型は，基本型および出力能力のみからなり，ALπ

の局所性を特徴付けている．型環境は Γで表され，Γの

Types

S, T ::= V value type

| L link type

| � behaviour type

Value types

V,W ::= B basic type

| oV output capability

Link types

L ::= iV input capability

| oV output capability

| �V both capability

Type environments

Γ ::= Γ, x : L

| Γ, x : V

| ∅
図 4 ALπ：型の文法

Fig. 4 ALπ: grammar of types.

もとでプロセス P が型付け可能であるとき Γ � P : �
と書く．

ALπ におけるプロセスの型付け規則およびサブタ

イプ規則は図 5 で与えられる13)．ここで，本論文で

は基本型の間にサブタイプ関係を定義しない．

π 計算のサブタイプ型システムでは，サブタイプ規

則は Sub-Trans，Sub-II，Sub-OO，Sub-BBを含

む．しかしながら，本論文では，通信リンクに用いら

れる値型が oo · · ·V の形のみであるため，Sub-trans

については必要ないため，本論文で用いる ALπ のサ

ブタイプ関係は Sub-refl，Sub-�I，Sub-�Oのみで

すべて導出できる．

Proposition 2.1 mV ≤ mW (m ∈ {i, o, �}) が，
規則Sub-refl，Sub-�I，Sub-�O，Sub-II，Sub-OO

と Sub-BBから導出されるとき，V = W．

Proof oV ≤ oW の場合のみ示す．V が基本型

のとき，W ≤ V は Sub-refl から導出されるため，

V = W．一方，V = oV ′ and W = oW ′ のとき，

oV ′ ≤ oW ′ は Sub-OO から導出される．帰納法の

仮定より，V ′ = W ′，よって V = W．iV ≤ iW と

�V ≤ �W の場合も同様． �

したがって，Sub-II，Sub-OO と Sub-BBで得ら

れるサブタイプ関係が Sub-Refl のみで得られるこ

とがいえる．この性質に基づいて，型クラスによって

サブタイプ関係を定義することができるようになる．

図 5 ALπ：型付け規則
Fig. 5 ALπ: typing rules.

3. 非同期 π 計算の Haskell でのコーディ
ング

本章で，サブタイプ関係がない非同期 π 計算（Aπ）

の Haskell での表現を示す．ここで問題となるのが，

ν 演算子に対応する一意な名前の生成方法である．

ここでは複数のモナドの機能の組合せにより，一意

な名前を生成する機構を持つ PiMonad を与え，Aπ

をHaskellで表現する．モナドの構造を把握するため，

まず，ν 演算子なしの非同期 π 計算（Aπ−ν）のプロ

セスの Haskellでの表現を与え，Aπ−ν の文脈（con-

text）を表す継続渡しモナド Ctx を与える．次に，名

前の生成器の状態をモナドに保持させるため，ν 演算

子を含んだ Aπ の文脈を，Ctx に状態計算を導入した

モナドである PiMonadで与える．

3.1 Aπ−ν のHaskellでの表現

チャネルの型定義 型 a を送受信する通信リンク

�aを表す型を，Ch aとして表す．たとえば，π 計算に

おいて型 �String で表されるチャネルを Ch String，

型 ��() を Ch (Ch ()) 等と表す．一方，名前には型

の情報は与えず，すべて整数で表す．チャネル型 Ch

a の定義を以下のように与える．

data Ch a = Name Int

型 Ch a は，定義の左辺に現れる型変数が右辺には

現れない幽霊型（phantom type）であり，型情報を

忘却することで，名前はすべて Int 型の整数表現を

Process type

data P =

Par P P parallel composition P |Q

| Zero inactive process 0

| forall t.(Typeable t) => Recv (Ch t) (t -> P) input x(y).P

| forall t.(Typeable t) => Send (Ch t) t asynchronous output xy.0

| forall t.(Typeable t) => ReprRecv (Ch t) (t -> P) replicated input !x(y).P

図 6 Aπ−ν の Haskell での表現
Fig. 6 Representation of Aπ−ν in Haskell.

表 1 Aπ−ν と Haskell の項の対応関係
Table 1 Correspondence between Aπ−ν and Haskell.

Aπ−ν Haskell

x(y).P Recv x (\y->p)

xv Send x v

!x(y).P ReprRecv x (\y->p)

P | Q Par p q

0 Zero

x(y).(yz|!z(w).wv) Recv x (\y->

Par (Send y z)

(ReprRecv z (\w->

Send w v)))

持つ一方，送受信のためには違った型を持つ．

プロセスの型定義 次に，Aπ−ν のプロセスの表現

を与える．Haskell で，出力プロセス xv を Send x

v，並行合成プロセス P |Qを Par p q，終了したプロ

セス 0 を Zero で表す．入力プロセス x(y).P による

名前束縛は λ 束縛で表現し，Recv x (\y->p) で表

す．複製 !x(y).P も同様に ReprRecv x (\y->p) と

表す．他に例として表 1 のような対応関係がある．

各データ構築子を，forall t. Send (Ch t) t や

forall t. Recv (Ch t) (t->P) 等と決める．これ

により，型 �t のチャネルで送受信可能な値の型が t

であることを，Haskellの型システムが保証する．デー

タ型の forall 構文は Haskellで一般的な拡張構文で

ある．

しかしながら，forall で限量した型はそれぞれの

値の出現で違った型として扱われるため，τ 遷移の計

算に必要な “Send c1 v と Recv c2 f から f v を計

算”といった記述は Haskellでは型付けできずコンパ

イルエラーとなる．

ここで，forall t. (Typeable t) => Send (Ch

t) t のように型 t が Typeable クラスに属すれば，

関数 cast による型キャストを行うことができる．こ

れにより，プロセスの τ 遷移を計算する関数を以下

のように定義する．

tau :: P -> P -> Maybe P

tau (Par (Send (Name i) v)

(Recv (Name j) f)) =

if i==j then Just (Par Zero (f (cast v)))

else Nothing

これらのデータ構築子をまとめ，チャネル型とあわ

せて，Aπ−ν を表す型の定義を図 6 で表す．チャネ

ル自身も送受信可能とするため，データ型 Ch a も

Typeable のインスタンスとして宣言する☆：

instance Typeable (Ch a) where

typeof = ...

3.2 Aπ−ν の文脈の継続モナドによる表現

モナドによるプロセスの表現は，後の名前生成を導

入した体系において不可欠な役割を果たす．ここでは，

まず型 P を結果の型とする継続モナド Ctx を次のよ

うに定義する：

data Ctx a = Ctx {runCtx :: ((a -> P) -> P)}

instance Monad Ctx where

return x = Ctx (\k -> k x)

(Ctx c) >>= f =

Ctx (\k -> c (\a -> runCtx (f a) k))

型 Ctx a を持つモナドは型 a を持つある変数を束

縛する文脈と見なすことができる．>>= 演算子は，あ

る文脈と同じ型の自由変数を持つ別の文脈から新たな

文脈を合成する演算と見なせる．

文脈の穴にプロセス 0を代入しプロセスを得る関数

ctx2pr を以下に定義する：

ctx2pr :: Ctx () -> P

ctx2pr (Ctx f) = f (\_ -> Zero)

さらに，モナド Ctx の組合せにより Aπ−ν の基本

的な演算子を表現するため，Aπ−ν の文脈に対応する

関数を与える．

入力 π 計算の入力ガード式 x(y).P から，文脈

x(y).[·]を得る．対応する関数 recvを以下に定義する：

☆ 現行の Hugs 98 March 2005 の場合．GHC6.4 では型構築
子 Ch を型 Typeable1 のインスタンスとして宣言するか，デー
タ型定義で deriving Typeable とする．

Generator for fresh names

type Gen = [Int]

initial = [1..] infinite list of integer

Unrestricted process

data APiSub =

Par APi APi parallel composition P |Q

| Zero inactive process 0

| forall x.(Typeable x) =>

Recv (Ch x) (x -> Gen -> (APi, Gen)) input x(y).P

| forall x.(Typeable x) =>

Send (Ch x) x asynchronous output xy.0

| forall x.(Typeable x) =>

ReprRecv (Ch x) (x -> Gen -> (APi, Gen)) replicated input !x(y).P

Process type

type APi = ([NewChan], APiSub) process with restriction

Restricted name

data NewChan = forall x. NewChan (Ch x) placeholder of the name

図 7 Aπ の Haskell での表現
Fig. 7 Representation of Aπ in Haskell.

recv :: (Typeable a) => Ch a -> Ctx a

recv ch = Ctx (\k -> Recv ch k)

引数は入力に用いられるチャネルを表す．定義におけ

る継続 k は型 a -> P を持ち，入力動作に継続する

プロセスを表す．

非同期出力 非同期出力プロセス xv.0 には継続す

るプロセスはない．継続するプロセスを配置するため

に変数束縛のない文脈（xv.0|[·]）を得る．対応する関
数 send を以下に定義する：

send :: (Typeable a) => Ch a -> a -> Ctx ()

send ch v = Ctx (\k ->

(Par (Send ch v) (k ())))

第 1引数は出力に用いられるチャネルを，第 2引数は出

力される値を表す．定義における継続 kは型 () -> P

を持ち，並行合成演算の右手側に置かれる．

並行合成 並行合成プロセス（P |P ′）より文脈

（P |[·]）を得る．対応する関数 forkを以下に定義する：

fork :: Ctx () -> Ctx ()

fork c =

let p = ctx2pr c

in Ctx (\k -> Par p (k ()))

第 1引数は別の文脈で，関数 ctx2pr によりプロセス

に変換された後，並行合成演算の左手側に置かれる．

複製 複製プロセス !x(y).P から，入力プロセスの

場合と同様に文脈 !x(y).[·] が得られる．しかしなが
ら，入力プロセスの場合と違い，継続するプロセスが

無限に複製されるため直観にあわない．そこで非同期

出力プロセスの場合と同様に，文脈（!x(y).P |[·]）か
ら対応する関数 rep を以下に定義する：

rep :: (Typeable a) =>

Ch a -> (a -> Ctx ()) -> Ctx ()

rep ch cont =

let k’ = \a -> ctx2pr (cont a)

in Ctx (\k ->

(Par (ReprRecv ch k’) (k ())))

第 1引数は入力に用いられるチャネルを，第 2引数は

入力動作に継続する文脈を表す．文脈は fork の場合

と同様に ctx2pr によりプロセスに変換され並行合成

される．

3.3 Aπ のHaskellでの表現

名前制限演算子を含む Aπ のプロセスの Haskellで

の表現を図 7 で与える．以下で，この直観的意味を説

明する．

名前生成器 名前は型 Int の整数値で表現する．ν

演算子で新しい名前を生成するための源として，Int

の十分に大きいリストを与える．これを名前生成器と

表 2 Aπ と Haskell の項の対応関係
Table 2 Correspondence between Aπ and Haskell.

Aπ Haskell

(νz : L)x(y).P ([NewChan z], Recv x

(\y->\gen->(([],p),gen)))

(νv : L)xv ([NewChan v], Send x v)

!x(y).(νz : L)P ([], ReprRecv x

(\y->\gen->(([NewChan z],p),gen)))

呼ぶ．名前生成器の型を Gen と名付ける．新しい名前

を表す整数はリストの head を用い，生成器の次状態

は tail を用いる．生成器の初期状態を initial と

名付ける．

プロセスの型 名前制限されたプロセスの Haskell

での表現を型 APi で与える．直観的に，そのスコー

プでの名前制限の（空かもしれない）リストと，名前

制限の下のプロセスの対である．

名前制限 1つの名前制限（νz: L）を，NewChan z

として表す．

名前制限演算子の下のプロセス 名前制限演算子の

下のプロセスを表す型を APiSub とする．データは

型 P にほぼ対応するが，入力プレフィクスに関わる

データ構築子 Recv，ReprRecv は，入力動作後に継

続が適用されプレフィクスの下のプロセスが活性化さ

れるため，引数に生成器を追加する必要がある．さら

に，適用した生成器の次の状態を得る必要から，継続

の戻り値にも生成器を追加する．よって，Recv およ

び ReprRecv を次のように定義する：

...

| Recv (Ch a) (a -> Gen -> (APi, Gen))

| ReprRecv (Ch a) (a -> Gen -> (APi, Gen))

...

継続するプロセスがない Send，Par，Zero につい

ては P のデータ構築子をそのまま用いる．

例 (νz : �a)xz を ([NewChan z], Send x z) と

表す．継続するプロセスで名前制限があるプロセス

x(y).(νz : �a)yz は，

Recv x (\y -> gen ->

let z = (Name (head gen)) :: Ch a

in (([NewChan z], Send y z), tail gen)

と表す．その他のプロセスの対応関係の例を表 2 にあ

げる．

3.4 状態モナドによる名前生成

ここで，Aπ の文脈に対応する一意な名前生成が可

能なモナドを，3.2節で与えた Aπ−ν の文脈に対応す

るモナドに，状態をともなう計算を表すモナドの機能

を組み合わせ構成する．

モナドで生成器の状態を保持するため，3.2節で導

newtype PiMonad a =

PiMonad {runPi::(a -> Gen -> (APi,Gen))

-> Gen

-> (APi,Gen)}

instance Monad PiMonad where

return x = PiMonad (\k -> \g -> k x g)

m >>= f =

PiMonad (\k -> g ->

runPi m (\a-> \g’ ->

runPi (f a) k g’) g

図 8 PiMonad：Aπ の文脈の Haskell での表現
Fig. 8 PiMonad: representation of Aπ context in Haskell.

入した継続モナド Ctx a に，状態モナドを合成する．

継続モナドと状態モナドの合成 CS を以下に示す：

data ContState a =

CS ((a-> S -> (R, S)) -> S -> (R, S))

instance Monad ContState where

return a = CS (\k -> \s -> k a s)

(CS f) >>= g =

CS (\k -> \s ->

f (\a -> \s’ ->

let (CS g’) = g a in g’ k s’)

s)

ここで状態を表す型を S，継続の結果の型を Rとしてい

る．直観的には，このモナドに型 (a -> S -> (R, S))

を持つ継続と，型 S を持つ初期状態を適用すると，型

(R, S) で表される継続の結果の型と最終状態の対が

得られる．

このモナドの状態の型を Gen，結果の型を APi と

して，Aπ の文脈を表すモナドを図 8 に示す．これを

PiMonadと呼ぶことにする．

本節の目的としていた名前制限の文脈 (νz : L)[·]
に対応する関数 new および補助関数 addNew を以下

に定義する：

new :: (Typeable a) => PiMonad (Ch a)

new = PiMonad (\k -> \gen ->

let newname = Ch (head gen)

gen’ = tail gen

(process, gen’’) = k newname gen’

in (addNew newname process, gen’’)

)

where addNew n (ns, p) = (NewChan n:ns, p)

新しい名前は生成器の head からとり，生成器の次状

態を tail としている．継続に新しい名前と生成器を

適用し，プロセスを得，名前制限を関数 addNew によ

りプロセスに記録する．

3.2節で定義した，Aπ−ν の基本的な演算子に対応す

る send，recv，rep，fork も，Aπ を表す PiMonad

の形に変更する必要がある．ここでは send，recv，

forkのみ示す：

send :: (Typeable a) =>

Ch a -> a -> PiMonad ()

send ch v = PiMonad (\k -> \gen ->

let (p, gen’) = k () gen

in (([], Par (Send ch v) p), gen’)

)

recv :: (Typeable a) => Ch a -> PiMonad a

recv ch = PiMonad (\k -> \gen ->

(([], Recv ch k), gen))

fork :: PiMonad () -> PiMonad ()

fork (PiMonad f) = PiMonad (\k -> \gen ->

let (p, gen’) = f (\_->Zero) gen

(q, gen’’) = k () gen’

in (([], Par p q), gen’’)

)

send において，継続 k に生成器 gen を適用し並行

合成するプロセス p を得る．一方，recv は生成器を

消費しない．入力動作が行われるときに生成器が適用

され，名前生成が行われる．rep も同様に定義される．

3.5 抽象評価器による PiMonadの実行

PiMonad () 型の値として定義したプロセスはそ

のままでは実行されることはない．Aπ の項と同様

な実行意味を与えるのは，Haskell で定義した関数

eval :: APi -> [Trans] である．eval の 1 回の

適用により 1ステップ縮約され，Aπ の遷移規則に対

応した動作をする．ここで，Trans は Aπ における遷

移ラベルと遷移後のプロセスの対 (l, P ) に相当する．

eval は図 2 の遷移規則に対応した実行として定義さ

れているものとする．

遷移を表す型 Trans の定義 Trans を，以下のよ

うに定義する：

data Trans =

TauAct APi

| forall x. (Typeable x) =>

SendAct [NewChan] (Ch x) x APi

| forall x. (Typeable x) =>

RecvAct (Ch x) (x -> APi)

データ構築子 TauAct は，規則 Comm や Close-

{L,R}に起因する τ 遷移に相当する．引数は，遷移

後のプロセスに相当する．

SendAct は，規則 AOutに起因する出力遷移に相

当する．第 2引数と第 3引数は，それぞれ遷移ラベル

のチャネルと送られる値に対応する．第 4引数は，遷

移後のプロセスに対応する．第 1引数（[NewChan]）

は，送られる名前がチャネル型の場合に束縛された名

前のリストに対応する．第 1 引数が空でないときは，

規則 Openに対応する実行と見なせる．

RecvAct は，規則 Inpまたは RepInpに起因する

入力遷移に相当する．第 1引数は遷移ラベルのチャネ

ルに相当する．第 2引数は，入力値に適用すると遷移

後のプロセスを返す関数を表す．

PiMonadの実行 PiMonadの実行は，プロセス

を関数 eval の適用結果から 1 つのプロセスを選び，

再度，eval を適用することを繰り返して得られる系

列となる．本論文における意味論は Early意味論であ

るため，入力遷移 RecvAct の場合は，関数から継続す

るプロセスを得るために外部からの入力が必要である．

4. i/o型とサブタイプ関係の導入

本章で，ALπ の 3つのチャネル型構築子 �，i，oに

対応する型を前章で導入したチャネル型から区別し，

さらに値型の概念を型クラスで定める．次に，図 5 で

紹介した ALπ のサブタイプ関係を Haskellの多引数

型クラス15) により表現する方法を示す．前章で導入

した send，recv，rep，new をこの型クラスを用い

た形で修正し，Haskell への ALπ の埋め込みを達成

する．

4.1 i/o型および値型の導入

ALπ の 3つのチャネル型と Haskellの関係をそれ

ぞれ，入力能力 ia を I a，出力能力 oa を O a，そし

て入出力チャネル �a を L a として与える．

ALπ では基本型および出力能力のみが値型である．

値型に対応する型クラス ValueType を定義し，出力能

力と，値として送受信可能としたい Haskellの基本型

を ValueType のインスタンスとして宣言する．cast

関数のため（3.1節），ValueType は Typeable のサ

ブクラスとして定義する．

上述 3つのチャネル型および値型を表す型クラスを

図 9 に与える．これを用い，サブタイプ関係を用い

ない ALπ を図 10 に与える．ここで，Aπ に対応す

る図 7 の型 APiSub のデータ構築子 Send (Ch x) x，

Recv (Ch x) · · ·，ReprRecv (Ch x) · · ·をそれぞれ
i/o型のために特化した型 ALPiSubの Send (O x) x，

Recv (I x) · · ·，ReprRecv (I x) · · · とし，さらに
型 x の動く範囲を ValueType のインスタンスに制限

している．名前制限のために用いられる NewChan は

Channel for both input and output capability

data L a = L Integer

Input capable channel

data I a = I Integer

Output capable channel

data O a = O Integer

Type class for value types

class (Typeable a) => ValueType a where

-- no functions

instance ValueType Int where

instance ValueType Float where

...

instance (ValueType a) => ValueType (O a) where

図 9 i/o チャネル型と値型の Haskell での表現
Fig. 9 Refined channel types and value types in PiMonad.

Unrestricted process

data ALPiSub =

Par ALPi ALPi parallel composition P |Q

| Zero inactive process 0

| forall x.(ValueType x) =>

Recv (I x) (x -> Gen -> (ALPi, Gen)) input x(y).P

| forall x.(ValueType x) =>

Send (O x) x asynchronous output xy.0

| forall x.(ValueType x) =>

ReprRecv (I x) (x -> Gen -> (ALPi, Gen)) replicated input !x(y).P

Process type

type ALPi = ([NewChan], ALPiSub) process with restriction

Restricted name

data NewChan = forall x. NewChan (L x) placeholder of the name

図 10 ALπ の Haskell での表現
Fig. 10 Representation of ALπ in Haskell.

チャネル型 L に特化する．

4.2 サブタイプ関係の型クラスによる表現

サブタイプ関係を型付けに関連付ける SubSump-

tion規則を，coercion 11)の役割を果たすメソッドを

持つ多引数型クラスで表現する．

出力チャネル型 出力チャネルには，型 �V または

oV を用いる．�V は oV に coerceする．これを次の

多引数型クラス OChan で表す：

class (ValueType a) => OChan o a where

coerceO :: o a -> O a

instance (ValueType a) => OChan O a where

coerceO = id

instance (ValueType a) => OChan L a where

coerceO (L i) = O i

出力対象の型はクラス宣言の文脈で値型に制限する．

メソッド coerceO は最外の型構築子を O に変換する．

インスタンスは O および Lについて宣言され，それぞ

れの coerceO の実装は，恒等関数と，名前表現を保

持し型のみを L から O に変換する関数で与えられる．

入力チャネル型 ALπ では，入力チャネルには，型

�V または iV を用いる．出力チャネルと同様に，次

のとおり型クラス IChan を宣言すれば十分である：

class (ValueType a) => IChan i a where

coerceI :: i a -> I a

instance (ValueType a) => IChan I a where

coerceI = id

instance (ValueType a) => IChan L a where

coerceI (L i) = I i

値型 値型についてはサブタイプ関係を直接多引数

型クラスと coercion のための関数を与える．次の型

クラス Subtype によりこれを表現する：

class (ValueType c2) =>

Subtype c1 c2 where

coerce :: c1 -> c2

instance (ValueType v) =>

Subtype (L v) (O v) where

coerce (L i) = O i

instance (ValueType v) =>

Subtype (O v) (O v) where

coerce = id

instance Subtype Int Int where

coerce = id

instance Subtype Float Float where

coerce = id

(and many declarations for other base types)

スーパタイプが値型であることを要請するため，クラ

ス宣言の文脈で型の第 2引数を ValueType に制限す

る．図 5 に現れるサブタイプ関係のうち，Sub-�Oに

対応する coercion関数のみ，型のみを L から O に変

換する関数で与える．残りはすべて Sub-Refl に対

応するため，恒等関数で与える．

ALπ の演算子への適用 導入した型クラスを用い

て，3章で導入した基本的な演算子 send，recv を修

正し図 11 に示す．rep も同様に与える．new，fork

についてはサブタイプ関係と関連しないため省略する．

これらの関数により，ALπ のプロセスが Haskell に

埋め込まれる．

例 4.1 PiMonad によるセマフォの記述 図 12

send :: (OChan c a, Subtype b a)

=> c a -> b -> PiMonad ()

send ch v = PiMonad (\cont -> \gen ->

let (cont’,gen’) = cont () gen

ch’ = coerceO ch

v’ = coerce v

p = ([], Send ch’ v’)

in (([], Par p cont’), gen’)

)

recv :: (IChan i a) => i a -> PiMonad a

recv ch = PiMonad (\f -> \gen ->

let ch’ = coerceI ch

in (([], Recv ch’ f), gen)

)

図 11 サブタイプ関係を表現した型クラスの ALπ への適用
Fig. 11 Use of defined type classes in ALπ operators.

に，2.2節のセマフォを PiMonadによって記述する．

ここで send sync x は同期のためのプロセス抽象で，

x(v)v() に対応する．セマフォの最大値の指定に変数

cnt を用い，標準ライブラリ関数 Control. Monad.

replicateM でプロセス send a ()を複製している．

5. Haskellの I/Oシステムの統合

本章では，PiMonadを Haskellの I/Oシステムと

統合する手法について示す．Haskell の IO アクショ

ンは，ALπ の出力動作として定式化する．

まず，ALπ の出力能力型に，Haskellの IOアクショ

ンを対応付ける拡張を行う．次に，I/Oシステムから

値を返却するよう拡張することで，I/Oシステムとの

相互作用を可能にする．

5.1 1方向通信

Haskellの I/Oシステムとの相互作用の実装として，

まずは値を与えて出力操作を実行するのみの 1方向の

通信を実装する．

Haskellの入出力は，IO モナドで記述したアクショ

ンを介して行われる（π 計算のアクションと区別する

ため Haskellアクションと呼ぶことにする）．Haskell

アクションを出力チャネルとして記述するため，チャ

ネル型 O を以下のように修正する：

data O a =

O Integer

| Action String (a->IO ())

以後，データ構築子 O のチャネルをローカルチャネ

ル，Action のチャネルをアクションチャネルと呼ぶ．

send_sync :: (ValueType a) => O (O a) -> PiMonad a

send_sync ch = do

(a::L a) <- new

send ch a

recv a

semaphore :: (L (O (O (O ()), O (O ())))) -> Int -> PiMonad ()

semaphore sem cnt = rep sem $ \x -> do

(p::L (O ())) <- new

(v::L (O ())) <- new

send x (p, v)

(a::L ()) <- new

rep a $ \_ -> do

r <- recv p

send r ()

s <- recv v

send s ()

send a ()

replicateM_ cnt (send a ())

user :: (L (O (O (O ()), O (O ())))) -> PiMonad ()

user sem = do

(x::L (O (O ()), O (O ()))) <- new

send sem x

(p, v) <- recv x

send_sync p

-- some works here --

send_sync v

all :: PiMonad ()

all = do

(sem::(L (O (O (O ()), O (O ()))))) <- new

semaphore sem 1 -- binary semaphore

user sem

図 12 PiMonad でのセマフォの記述
Fig. 12 Semaphore in PiMonad.

データ構築子 Actionの第 1引数は，系内で一意なチャ

ネルの名前を文字列で与える．第 2引数は，チャネルに

対する出力動作に対応する Haskellアクションを与え

る．アクションチャネルに対し出力動作を行うとき，こ

のHaskellアクションを関数 Control. Concurrent.

forkIO により並行実行するよう，3.5 節で導入した

評価器を拡張する．

例 5.1標準出力への非同期書き込み Haskellアク

ションの putStrLn を用いて，標準出力へのチャネル

cout を作る：

cout :: O String

cout = Action "cout" (\x -> putStrLn x)

cout を用いて，標準出力へ文字列 "Hello, world!"

の書き込みを行うプロセスを以下のように定義できる：

hello :: PiMonad ()

hello = send cout "Hello, World!"

5.2 同期と双方向通信

Haskellの I/Oシステムからの値の返却 ここま

fopenW :: O (String, O (O (String,O ()), O (O ())))

fopenW = Action "fopenW" (\(filename,retCh) -> do

handle <- openFile filename WriteMode

let write :: O (String, O ())

write = Action ("write"++show handle) (\(str,ch) -> do

hPutStrLn handle str

putBuf ch ()

)

close :: O (O ())

close = Action ("close"++show handle) (\ch -> do

hClose handle >> putBuf ch ()

putBuf ch ()

)

putBuf retCh (write, close)

)

where

putBuf :: O a -> a -> IO ()

putBuf (Action _ f) v = f v

図 13 ファイル書き込みチャネル fopenW

Fig. 13 File writer channel fopenW.

での実装では，非同期出力に対応するHaskellアクショ

ンが発火しても，Haskellアクションから評価器への

値の返却の手段が存在しないため，Haskellアクショ

ンの完了を待機するプロセスや，戻り値を持つような

Haskellアクションの呼び出しを記述することができ

ない．

Haskellの I/Oシステムが返却値を格納し，評価器

が取り出すバッファ RecvBuf を，FIFOバッファであ

る Control. Concurrent. Chan を用いて準備する：

type RecvBuf = Chan RecvData

data RecvData =

forall x. (ValueType x) =>

RecvData (O x) x

RecvDataは受信に用いたローカルチャネルと受け取っ

た値を引数に持つ．バッファに投入された RecvData

型のデータは，評価器により取り出され，出力プロセ

スとして PiMonadのプロセスと並行合成される．こ

うして値の受け渡しが実現する．たとえばアクション

チャネル x を介して値 v が送られた場合，バッファ

にはデータ RecvData x v が投入され，評価器によ

りプロセス Send x v が並行合成される．

次に，Haskellアクションが値を RecvBuf に格納す

るための方法を述べる．

Haskell アクション内でのローカルチャネルの扱

い PiMonadの評価器から Haskellの I/Oシステム

へローカルチャネルを渡す際，渡すローカルチャネル

を，RecvBuf へ値を投入するアクションチャネルへ

変換する．Haskellアクション側は，これを実行する

ことで，値の返却を行う．このため，前章で導入した

ValueType に，以下のように変換メソッド trans を

追加する：

class (Typeable x) => ValueType x where

...

trans :: RecvBuf -> x -> IO x

trans _ x = return x

trans は変換を行わないデフォルトメソッドを持つ．

出力チャネルについてのみこれをオーバライドする：

instance (ValueType x) =>

ValueType (O x) where

...

trans buf ch@(O i) =

return (Action (show i) (\x ->

writeChan buf (RecvData ch x)))

trans _ ch = return ch

trans は，ローカルチャネルを，RecvBuf へ値を投入

（writeChan）するアクションチャネルへ変換する．ア

クションチャネルの文字列表現には元のローカルチャ

ネルの整数表現が用いられる（show i）．

例 5.2ファイル書き込みチャネル ファイルへの書

き込みに用いるチャネル fopenW を図 13 に定義する．

fopenW はファイル名を表す文字列 filename とチャ

ネル retCh の対を受け取り，ファイルを開き，書き

込み用チャネル write とファイルを閉じるためのチャ

ネル close の 2つを，渡されたチャネル retCh を介

して返す．書き込み用チャネルは書き込む文字列 str

とチャネル ch を受け取る．文字列がファイルに書き

込まれた後，ch を介し値 () を送信する．これにより

利用者は書き込みの完了を知ることができる．close

も同期のためのチャネルを受け取る．

fopenW を使い，ファイル書き込みは以下のように

書くことができる：

helloWriter :: String -> PiMonad ()

helloWriter filename = do

(ret::O (O (String, O ()), O (O ())))

<- new

send fopenW (filename, ret)

(write,close) <- recv ret

(sync::O ()) <- new

send write ("Hello world!",sync)

_ <- recv sync

send_sync close

6. ネットワーク実装

本章では，Haskellで提供されるネットワーク API

と，前章で与えた Haskell の I/O システムへの拡張

を組み合わせて，PiMonadのネットワーク実装を与

える．

表面上は 2つのチャネルを加えることで実現できる．

1つは，リモートホストからの接続を待ち受けるチャ

ネルであり，もう 1つはリモートホストへと接続する

ためのチャネルである．これらを初期チャネルと呼ぶ

ことにする．通信するホストどうしは，まずこれら 2

つのチャネルを介し相互にローカルチャネルを渡しあ

う．交換したチャネルを介して，通信の続きを行う．

さらに，前章で定義した出力チャネルに，ネットワー

ク上の通信ホストの位置情報を含めるよう修正する．

他に，Haskellのネットワーク通信に必要なスレッド

を準備し，値の直列化を実装する．

6.1 基本メカニズム

直列化 Haskellのネットワーク通信は文字列の通

信として記述されるため，値型として定義した型クラ

ス ValueType に文字列への直列化と，値の復元の関

数を追加する．通信は RecvBuf を直列化し行う．す

なわち，各通信ごとに，ピア間で送信チャネルと値の

文字列表現の対がやりとりされる．

アクションチャネルの修正と接続ホスト管理 アク

ションチャネルを，ネットワーク接続情報を追加する

ために以下のように修正する：

data O a =

O Integer

| Action

String

Location

(PiPeer -> a -> IO ())

追加された第 2引数 Location は，ホスト名等のネッ

トワーク上の通信ホストの位置情報を格納するため

に用いる．Haskellアクションの追加引数 PiPeer は，

(1) 当該ホストのホスト名，(2) 受信のために用いら

れるローカルチャネルと，その文字列表現を対応付け

るテーブル，(3) 利用できる TCPコネクションのリ

ストからなるデータである．

着信モニタ 評価器に TCPコネクションの待ち受

けスレッド（モニタ）を追加する．コネクション要求

を受け付けたとき，モニタは受信データを待ち受ける

スレッド，送信のためのスレッドを 1つずつ生成する．

2つのスレッドは，それぞれ入力動作と出力動作に対

応した通信を行う．

初期入力チャネル 初期チャネルのうちの 1 つが，

初期入力チャネルである．初期入力チャネルは TCP

コネクション確立時に入力動作が発火される．すべて

のホストは，後で導入する名前解決チャネルを介して

お互いの初期入力チャネルを得，このチャネルに向け

て送信を行い，コネクションを確立する．初期入力チャ

ネルはトップレベルの PiMonadプログラムの引数と

してのみ与えられる．初期入力チャネルは型 I a を

持ち，型変数 a はアプリケーション固有の通信の内容

を表すように使用者側で具体化する．

例 6.1エコーバック・サーバ 初期入力チャネルで

着信を待ち，文字列とチャネルの対を受け取り，チャ

ネルを介して文字列をそのまま返すエコーバック・サー

バの例を示す：

echo :: I (String, O String) -> PiMonad ()

echo init =

rep init (\(str, reply) -> send reply str)

ここで引数 init が初期入力チャネルである．多相的

な型付けにより，初期入力チャネルで受け取る型は任

意の型をとることができる．

名前解決チャネル もう 1つの初期チャネルが，名

前解決チャネル resolve である．名前解決チャネルは

ホスト名とポート番号の対を受け取り，対応するホス

トで待ち受けているプロセスの初期入力チャネルのコ

ピーを返す．このチャネルを介した出力動作が，TCP

コネクションの確立に対応する．resolve は次の型シ

グネチャを持つ：

resolve :: (ValueType a) =>

O ((Hostname,PortNumber), O (O a))

型変数 a は，初期入力チャネル同様，使用者側で具体

化する．

ネットワーク通信における各チャネル型の役割 初

期入力チャネルはリモートホストからのコネクション

を待つためだけに用いるため，入力能力型 I a を持

つ．一方，通信において多く用いるのは出力能力型 O

a とチャネル型 L a である．ALπ の局所性のため，

他プロセスから入力プレフィクスで受信するチャネル

はすべて出力能力型である．出力能力型はネットワー

クの通信ホストの位置情報を持つよう本章で拡張する．

一方，new で生成するチャネルはチャネル型を持ち，

send により出力能力のみをリモートホストに渡しそ

のチャネルで入力することで，継続したネットワーク

通信が可能になる．

6.2 多相型付けされた初期チャネルと通信の安全性

アプリケーション固有の型を送受信する余地を残す

ため，2つの初期チャネルは多相的に型付けされてい

る．それゆえ，それぞれのホストで期待する値型が違

う可能性があり，通信が失敗するおそれがある．

そこで，初期チャネルでの通信のみ，相互の型の比

較を型名のレベルで行う．初期チャネルの型が整合し

ていれば，継続する通信の型はつねに整合していると

推測される．

6.3 問 題 点

名前の枯渇 π 計算は加算無限個の名前の集合を仮

定している．本フレームワークにおいて名前はHaskell

処理系の実装に依存した有限の整数で表現されるため，

サーバプロセス等長時間動作するプロセスでは名前の

数が足りなくなることがある．

振舞い等価性等の諸規則により使われない名前を特

定してごみ集めすることはある程度可能である．しか

しながら，I/Oシステムやリモートホストに送信され

た名前についての参照カウント等を用いて管理しなけ

れば，名前の枯渇を防ぐことはできない．

プロセスの溢れ 同様に，それ以上動作しないプロ

セスがメモリ領域を占有することがある．そうしたプ

ロセスは π 計算の振舞い等価性により削除できる．た

とえば，(νx : L)(!x(y).P ) ∼=c 0 より，通信プロセ

ス ([NewChan x], ReprRecv x (\y->p)) はこれ以

上相互作用しないことが保証されるため，系から取り

除くことができる．

7. プログラム例

本章では，構築したフレームワークを用いてインス

タント・メッセンジャの例を示す．プログラムはクラ

イアントがサーバに接続することでメッセージをやり

とりする．キーボードや画面表示は単一チャネルで表

されているものとし，ネットワーク通信に関係のある

部分のみ取り上げる．

7.1 メッセージ交換プロセス

図 14 で与えられるプロセス imMain はサーバとク

ライアントの両方で実行される．このプロセスはメッ

セージ交換に共通する部分を与える．最初の 2 つの

引数はそれぞれユーザへの入力と出力を表すものとす

る．残りの 2つの引数は，リモートのユーザ入力と出

力である．imMain は並列に起動される receiver と

sender によって成り立つ．

receiverはループ用チャネル rloopで待った後，リ

モートホストからチャネル rinを介し入力を待つ．メッ

セージが来ると，ただちにユーザに送られ，メッセー

ジが表示されるまで待つ．表示が完了したら，rloop

に値 () を送り次のサイクルを起動する．

senderはループ用チャネル sloopで待った後，ユー

ザ入力へチャネルを送り，待つ．ユーザ入力がなされ

た時点でリモートホストへ入力文字列を送る．この際

sloop をリモートホストに送ることで，リモートホス

トからの応答があった時点でただちに次のサイクルが

起動される．

7.2 スタートアッププロセス

図 15 の server，client はそれぞれサーバおよ

びクライアントのスタートアッププロセスである．そ

れぞれ初期チャネルを用いて通信を開始する部分であ

る．サーバ，クライアントともに仮引数 cin，仮引数

cout はそれぞれローカルのユーザ入力と出力に対応

するものとする．

server は，初期入力チャネル incoming よりクラ

イアントへメッセージを送るチャネル rout と，クラ

イアントからのメッセージを受け取るチャネルを送る

チャネル orin の対を受け取る．ユーザにコネクショ

ンが来た旨を伝えた（"connection comes"）後，ク

ライアントからのメッセージを受け取るチャネル rin

を受け取り，imMain へと継続する．

client は，引数 hostname および portnumber で

与えられるホストの初期チャネルを，名前解決チャネ

ル resolve1 を介して送ったチャネル res より取得す

る．resolve1 は多相的に型付けされた resolve チャ

ネルを具体型にしている．初期チャネルが取得できた

imMain ::

(I (String, O ())) -> (O (String, O ())) ->

(L (String, O ())) -> (O (String, O ())) -> PiMonad ()

imMain cin cout rin rout = do

(rloop::L ()) <- new

rep rloop (receiver rloop)

(sloop::L ()) <- new

rep sloop (sender sloop)

send rloop ()

send sloop ()

where

receiver :: L () -> () -> PiMonad ()

receiver rloop _ = do

(str, sync) <- recv rin

(sync::L ()) <- new

send cout (str, sync)

_ <- recv sync

send rloop ()

sender :: L () -> () -> PiMonad ()

sender sloop _ = do

(inp::L String) <- new

send cin inp

str <- recv inp

send rout (str, sloop)

図 14 メッセージ交換プロセス imMain

Fig. 14 imMain: the core of the IM program.

とき，ユーザに文字列 "connected." を通知し，チャ

ネルのやりとりによりリモートからの受信チャネル

rin，リモートへの送信チャネル rout を得て imMain

へと継続する．

7.3 従来のHaskellネットワークプログラムとの

比較

HaskellのネットワークAPIを用いた通信では，文

字列と値の相互変換を行う必要がある．通信手順の誤

りにより，たとえば，整数型の値を受信しようとして

いるピアに対し，文字列型の値を送信し，値の復元に

失敗し実行時エラーとなる場合がある．

本フレームワークでは，初期チャネルの型が一致し

ていれば，データの直列化と復元に失敗せず，後の通

信が失敗しないと期待できる．

8. 関 連 研 究

本章では，関連研究として，プロセス計算に基づく

プログラミング言語と，Haskellのモナドによる並行

プロセスの表現に関する研究をあげる．

8.1 プロセス計算に基づくプログラミング言語

π 計算に基づく既存のプログラミング言語としては，

Pict 12)，Join計算3)，Nomadic Pict 16)やNepi 7)が

知られている．Nomadic Pictは，Pictの拡張で分散

実装を持ち，π 計算を拡張した Nomadic π 計算に基

づき，モバイルエージェントの記述が可能である．Pict

では足し算や引き算等の基本的な演算をもチャネルと

して提供しているため，より細かく π 計算の代数法

則や遷移規則による解析が可能であると考えられる．

一方，PiMonadは Haskell の豊富な関数とデータ構

造を用いることができるという利点がある．

Join計算も非同期で局所的な性質を持つ計算に基づ

いている．Join計算は CHAMの動作に基づいてでき

るだけシンプルなメカニズムで柔軟な同期機構を統一

的に記述することを目的としている．これに対して，

PiMonadは，埋め込み言語として，Haskellの拡張と

して定義し，Haskellとネットワーク通信をシームレ

スで記述することを目的としている．

Nepi は Lisp 処理系上に実装された，同期 π 計算

server ::

O (O (String, O ())) -> O (String, O ()) ->

I (O (String, O ()), (O (O (String, O ())))) -> PiMonad ()

server cin cout incoming = do

(rout,orin) <- recv incoming

(sync::L ()) <- new

send cout ("connection comes.", sync)

_ <- recv sync

(rin::L (String, O ())) <- new

send orin rin

imMain cin cout rin rout

client :: String -> Int ->

O (O (String, O ())) -> O (String, O ()) ->

I () -> PiMonad ()

client hostname portnumber cin cout _ = do

(res::L (O (O (String, O ()), O (O (String, O()))))) <- new

send resolve1 ((hostname,portnumber), res)

remote <- recv res

(sync::O ()) <- new

send cout ("connected.", sync)

_ <- recv sync

(rin::L (String, O ())) <- new

(irout::L (O (String, O()))) <- new

send remote (rin, irout)

rout <- recv irout

imMain cin cout rin rout

resolve1 :: O ((String, Int),O (O (O (String, O ()), O (O (String, O())))))

resolve1 = resolve

図 15 スタートアッププロセス server および client

Fig. 15 server and client: the startup processes of IM program.

に基づくプログラミング言語である．関数型言語上に

構築された点において本研究との類似がある．しかし

ながら Lispは型システムを提供しないため，Nepiに

は型システムの概念がない．一方，Nepi は同期通信

や非決定的選択プリミティブを提供し，より制御構造

を見通し良く記述できるとしている．

8.2 モナドによる並行プロセスの表現

継続モナドにより並行モナドを構成する手法は文

献 2) や 14) でも述べられている．これらに対して，

我々の手法では並行プリミティブに非同期 π 計算の

ものを用い，チャネル型の概念による安全な通信が期

待できる．さらに，状態モナドを組み合わせることで

名前生成を可能にした．

9. まとめと今後の課題

本章では，まとめと今後の課題を述べる．

9.1 ま と め

本論文で，我々は非同期局所化 π 計算（ALπ）に

基づくネットワークプログラミングフレームワークと

して PiMonad および非同期局所化 π 計算に基づく

型付け手法を提案した．PiMonadを定義することで

Haskellに対する埋め込み言語としてALπ 計算の演算

子を実現した．ALπ では，型付けを行うためにサブタ

イプ関係が必要であるため，ALπ の型付けにおけるサ

ブタイプ関係の表現には Haskellの標準的な拡張であ

る多引数型クラスを用いた．本手法は単純な coercion

による定義ができない一般のサブタイプ関係につい

ては適用できない．本フレームワークでは，データに

基づくサブタイプ関係を仮定しないことで Haskellで

ALπ の型システムを表現することに成功した．

ALπの操作意味論に基づく評価器によって，Haskell

の IO アクションを π 計算のチャネルとしてモデル化

した．プログラマは任意のアクションをチャネルとし

て記述でき，それ自身も他のチャネルを介して通信で

きる．具体例としてファイル書き込みチャネルの例を

示した．さらにネットワーク通信の実装において，2

つの初期チャネルを用いたネットワークプログラミン

グを実現した．

例としてインスタント・メッセンジャの例を示した．

9.2 今後の課題

型付き ALπ 計算のラベル付き遷移関係と評価器に

よる PiMonadの実行の間には，直観的には双模倣の

対応があると予想される．しかしながら，異常終了等

の実際の評価器の振舞いが十分抽象的に形式化されて

いないため，検証は今後の課題とする．

この後さらに，型安全性を保証することが課題とし

てあげられる．たとえば，PiMonadで記述したプログ

ラムは，String 型とチャネル型の取り違え等に起因

する実行時エラーが生じない．このようなエラーは一

般のネットワーク通信プログラミングでは起こりうる

ため，静的な解析による検証は有益である．PiMonad

で導入した値の Haskell における型判定と，ALπ の

型判定の対応関係を定式化しこれを証明する．

一方，現在の実装では 6章であげたようなメモリ管

理上の問題点があるため，チャネルの生存管理を行う

必要がある．

他に，π 計算の規則を利用したプログラムのより詳

細な解析・検証があげられる．たとえば，文献 4)や 5)

の機密度を付加した型システムを導入し，機密性を保

証するフレームワークを構築することがあげられる．

